/*******************************************************************************
 * Copyright (c) 2000, 2004 IBM Corporation and others.
 * All rights reserved. This program and the accompanying materials
 * are made available under the terms of the Common Public License v1.0
 * which accompanies this distribution, and is available at
 * http://www.eclipse.org/legal/cpl-v10.html
 * 
 * Contributors:
 *     IBM Corporation - initial API and implementation
 *******************************************************************************/
module dwt.widgets.fontdialog;


private import dwt.dwt;

private import dwt.graphics.fontdata;
private import dwt.graphics.rgb;
private import dwt.internal.compatibility;
private import dwt.internal.win32.os;
private import dwt.util.util;
public import dwt.widgets.dialog;
private import dwt.widgets.display;
private import dwt.widgets.shell;


/**
 * Instances of this class allow the user to select a font
 * from all available fonts in the system.
 * <dl>
 * <dt><b>Styles:</b></dt>
 * <dd>(none)</dd>
 * <dt><b>Events:</b></dt>
 * <dd>(none)</dd>
 * </dl>
 * <p>
 * IMPORTANT: This class is intended to be subclassed <em>only</em>
 * within the DWT implementation.
 * </p>
 */
public class FontDialog : Dialog {

	FontData fontData;
	DWTRGB rgb;
	
/**
 * Constructs a new instance of this class given only its parent.
 *
 * @param parent a shell which will be the parent of the new instance
 *
 * @exception IllegalArgumentException <ul>
 *    <li>ERROR_NULL_ARGUMENT - if the parent is null</li>
 * </ul>
 * @exception SWTException <ul>
 *    <li>ERROR_THREAD_INVALID_ACCESS - if not called from the thread that created the parent</li>
 *    <li>ERROR_INVALID_SUBCLASS - if this class is not an allowed subclass</li>
 * </ul>
 */
public this (Shell parent) {
	this (parent, DWT.PRIMARY_MODAL);
}

/**
 * Constructs a new instance of this class given its parent
 * and a style value describing its behavior and appearance.
 * <p>
 * The style value is either one of the style constants defined in
 * class <code>DWT</code> which is applicable to instances of this
 * class, or must be built by <em>bitwise OR</em>'ing together 
 * (that is, using the <code>int</code> "|" operator) two or more
 * of those <code>DWT</code> style constants. The class description
 * lists the style constants that are applicable to the class.
 * Style bits are also inherited from superclasses.
 * </p>
 *
 * @param parent a shell which will be the parent of the new instance
 * @param style the style of dialog to construct
 *
 * @exception IllegalArgumentException <ul>
 *    <li>ERROR_NULL_ARGUMENT - if the parent is null</li>
 * </ul>
 * @exception SWTException <ul>
 *    <li>ERROR_THREAD_INVALID_ACCESS - if not called from the thread that created the parent</li>
 *    <li>ERROR_INVALID_SUBCLASS - if this class is not an allowed subclass</li>
 * </ul>
 */
public this (Shell parent, int style) {
	super (parent, style);
	checkSubclass ();
}

// added by <Shawn Liu>, to support center window to parent
extern (Windows)
static uint CFHookProc (HWND hdlg, uint msg, uint wParam, int lParam) {
	switch (msg) {
		case OS.WM_INITDIALOG:
			CHOOSEFONT* lpcf = cast(CHOOSEFONT*)lParam;
			FontDialog pThis = cast(FontDialog)cast(void*)lpcf.lCustData;
			// <Shawn Liu> center to parent
			OS.CenterWindow(hdlg, pThis.parent.getSafeHwnd());
			break;
		default : break;
	}
	return 0;
}

/**
 * Returns a FontData object describing the font that was
 * selected in the dialog, or null if none is available.
 * 
 * @return the FontData for the selected font, or null
 * @deprecated use #getFontList ()
 */
public FontData getFontData () {
	return fontData;
}

/**
 * Returns a FontData set describing the font that was
 * selected in the dialog, or null if none is available.
 * 
 * @return the FontData for the selected font, or null
 * @since 2.1.1
 */
public FontData [] getFontList () {
	if (fontData is null) return null;
	FontData [] result = new FontData [1];
	result [0] = fontData;
	return result;
}

/**
 * Returns the currently selected color in the receiver.
 *
 * @return the RGB value for the selected color, may be null
 *
 * @see PaletteData#getRGBs
 * 
 * @since 2.1
 */
public DWTRGB getRGB () {
	return rgb;
}

/**
 * Makes the dialog visible and brings it to the front
 * of the display.
 *
 * @return a FontData object describing the font that was selected,
 *         or null if the dialog was cancelled or an error occurred
 *
 * @exception SWTException <ul>
 *    <li>ERROR_WIDGET_DISPOSED - if the dialog has been disposed</li>
 *    <li>ERROR_THREAD_INVALID_ACCESS - if not called from the thread that created the dialog</li>
 * </ul>
 */
public FontData open () {
	version(WinCE) { 
		DWT.error(__FILE__, __LINE__, DWT.ERROR_NOT_IMPLEMENTED); 
		return;	
	}
	else
	{
	
	/* Get the owner int for the dialog */
	HWND hwndOwner = null;
	if (parent !is null) hwndOwner = parent.handle;

	/* Open the dialog */
	CHOOSEFONT lpcf;
	lpcf.lStructSize = CHOOSEFONT.sizeof;
	lpcf.hwndOwner = hwndOwner;
	lpcf.Flags = OS.CF_SCREENFONTS | OS.CF_EFFECTS | OS.CF_ENABLEHOOK;
	lpcf.lpfnHook = &CFHookProc;
	lpcf.lCustData = cast(DWORD)cast(void*)this;
	
	// you can't use struct here, this will pass to FontData
//	LOGFONT* lpLogFont = new LOGFONT();		
	LOGFONT lpLogFont;
//	if (fontData !is null && fontData.data !is null) {
	if (fontData !is null) {
		LOGFONT* logFont = &fontData.data;
		int lfHeight = logFont.lfHeight;
		HDC hDC = OS.GetDC (null);
		int pixels = -Compatibility.round (fontData.height * OS.GetDeviceCaps(hDC, OS.LOGPIXELSY), 72);
		OS.ReleaseDC (null, hDC);
		logFont.lfHeight = pixels;
		lpcf.Flags |= OS.CF_INITTOLOGFONTSTRUCT;
		OS.MoveMemory (&lpLogFont, logFont, LOGFONT.sizeof);
		// restore
		logFont.lfHeight = lfHeight;
	}
	lpcf.lpLogFont = &lpLogFont;
	
	if (rgb !is null) {
		int red = rgb.red & 0xFF;
		int green = (rgb.green << 8) & 0xFF00;
		int blue = (rgb.blue << 16) & 0xFF0000;
		lpcf.rgbColors = red | green | blue;
	}
	
	/* Make the parent shell be temporary modal */
	Shell oldModal = null;
	Display display = null;
	if ((style & (DWT.APPLICATION_MODAL | DWT.SYSTEM_MODAL)) != 0) {
		display = parent.getDisplay ();
		oldModal = display.getModalDialogShell ();
		display.setModalDialogShell (parent);
	}

	/* Open the dialog */
	boolean success = OS.ChooseFont (&lpcf);
	
	/* Clear the temporary dialog modal parent */
	if ((style & (DWT.APPLICATION_MODAL | DWT.SYSTEM_MODAL)) != 0) {
		display.setModalDialogShell (oldModal);
	}
	
	/* Compute the result */
	if (success) {

		/*
		 * This will not work on multiple screens or
		 * for printing. Should use DC for the proper device.
		 */
		HDC hDC = OS.GetDC(null);
		int logPixelsY = OS.GetDeviceCaps(hDC, OS.LOGPIXELSY);
		int pixels = 0; 
		if (lpLogFont.lfHeight > 0) {
			/*
			 * Feature in Windows. If the lfHeight of the LOGFONT structure
			 * is positive, the lfHeight measures the height of the entire
			 * cell, including internal leading, in logical units. Since the
			 * height of a font in points does not include the internal leading,
			 * we must subtract the internal leading, which requires a TEXTMETRIC,
			 * which in turn requires font creation.
			 */
			HFONT hFont = OS.CreateFontIndirect(&lpLogFont);
			HFONT oldFont = OS.SelectObject(hDC, hFont);
			TEXTMETRIC lptm;
			OS.GetTextMetrics(hDC, &lptm);
			OS.SelectObject(hDC, oldFont);
			OS.DeleteObject(hFont);
			pixels = lpLogFont.lfHeight - lptm.tmInternalLeading;
		} else {
			pixels = -lpLogFont.lfHeight;
		}
		OS.ReleaseDC(null, hDC);

		int points = Compatibility.round(pixels * 72, logPixelsY);
		fontData = FontData.win32_new (&lpLogFont, points);
		int red = lpcf.rgbColors & 0xFF;
		int green = (lpcf.rgbColors >> 8) & 0xFF;
		int blue = (lpcf.rgbColors >> 16) & 0xFF;
		rgb = new DWTRGB (red, green, blue);
	}
		
	/*
	* This code is intentionally commented.  On some
	* platforms, the owner window is repainted right
	* away when a dialog window exits.  This behavior
	* is currently unspecified.
	*/
//	if (hwndOwner != 0) OS.UpdateWindow (hwndOwner);
	
	if (!success) return null;
	return fontData;
	}
}

/**
 * Sets a FontData object describing the font to be
 * selected by default in the dialog, or null to let
 * the platform choose one.
 * 
 * @param fontData the FontData to use initially, or null
 * @deprecated use #setFontList (FontData [])
 */
public void setFontData (FontData fontData) {
	this.fontData = fontData;
}

/**
 * Sets a set of FontData objects describing the font to
 * be selected by default in the dialog, or null to let
 * the platform choose one.
 * 
 * @param fontData the set of FontData objects to use initially, or null
 * @since 2.1.1
 */
public void setFontList (FontData [] fontData) {
	if (fontData !is null && fontData.length > 0) {
		this.fontData = fontData [0];
	} else {
		this.fontData = null;
	}
}

/**
 * Sets the receiver's selected color to be the argument.
 *
 * @param rgb the new RGB value for the selected color, may be
 *        null to let the platform to select a default when
 *        open() is called
 *
 * @see PaletteData#getRGBs
 * 
 * @since 2.1
 */
public void setRGB (DWTRGB rgb) {
	this.rgb = rgb;
}

}
